

Terminale Ssv1 – Devoir N°11

Dans un repère orthonormal (O, \vec{i}, \vec{j}) , on considère le cercle (Ω) de centre O et de rayon 1 (la figure est à la fin du document).

On pose : $\overrightarrow{OH} = x \cdot \vec{i}$.

- 1/ Calculer l'aire du triangle MNI en fonction de x .
- 2/ On considère la fonction f définie sur $[-1; 1]$ par $f(x) = (1-x) \cdot \sqrt{1-x^2}$.
Étudier la dérivabilité de f en 1 et -1. On note \mathcal{C}_f la courbe représentative de f dans un repère orthonormal. En déduire une équation des tangentes à la courbe \mathcal{C}_f aux points d'abscisses 1 et -1.
- 3/ Étudier les variations de f sur $[-1; 1]$.
- 4/ Tracer la courbe \mathcal{C}_f .
- 5/ Pour quelle valeurs de x l'aire du triangle MNI est-elle maximale? Calculer alors cette aire.
- 6/ Démontrer qu'il existe un réel noté μ , autre que 0, qui vérifie $f(\mu) = 1$.
- 7/ Déterminer, à 10^{-2} près, une valeur approchée de μ par défaut.

1/ L'aire du triangle MIN est égale à $(1-x)\sqrt{1-x^2}$.

2/ Nous commençons par définir la fonction f au niveau de MAXIMA.

```
> f(x) := (1-x)*sqrt(1-x^2);
```

$$f(x) := (1-x) * \text{SQRT}(1-x^2);$$

Afin d'étudier la dérivabilité de f en 1 et -1 , nous allons définir le taux d'accroissement de f entre deux points x et y de façon générale.

$$> T(x, y) := (f(y) - f(x)) / (y - x);$$

$$T(x, y) := (f(y) - f(x)) / (y - x);$$

Calculons la limite de ce taux entre x et 1, lorsque x tend vers 1, à gauche.

$$> \text{limit}(T(x, 1), x, 1, \text{minus});$$

0

f est donc dérivable en 1 et sa courbe \mathcal{C}_f admet une tangente horizontale en ce point.

Procédons à la même étude locale, en -1 .

$$> \text{limit}(T(x, -1), x, -1, \text{plus});$$

 ∞

f n'est pas dérivable en -1 et \mathcal{C}_f admet une tangente verticale en ce point.

3/ Pour déterminer les variations de f sur $[-1, 1]$, il nous suffit d'étudier sa dérivée.

$$> a:\text{diff}(f(x), x);$$

$$-\sqrt{1-x^2} - \frac{(1-x)x}{\sqrt{1-x^2}}$$

Essayons de simplifier...

> a:radcan(a);

$$\frac{2x^2 - x - 1}{\sqrt{1-x}\sqrt{x+1}}$$

Nous allons noter g la dérivée de f .

> define(g(x),a);

$g(x) := (2*x^2 - x - 1) / (\text{SQRT}(1-x) * \text{SQRT}(x+1));$

Maintenant passons à la factorisation de g :

> factor(g(x));

$$\frac{(x-1)(2x+1)}{\sqrt{1-x}\sqrt{x+1}}$$

Les zéros de g sont 1 et $-\frac{1}{2}$, ce que nous pouvons retrouver avec MAXIMA.

> s:solve(g(x)=0);


$$\left[x = 1, x = -\frac{1}{2} \right]$$

Entre ces valeurs, la dérivée de f est négative; à l'extérieur elle est positive. Par ailleurs les valeurs atteintes par f sont, en ces points, respectivement :

> map(f,map(rhs,s));

$$\left[0, \frac{3\sqrt{3}}{4} \right]$$

4/ La représentation de f est la suivante :



- 5/ L'aire du triangle MNI est maximale pour $x = -\frac{1}{2}$ et elle vaut alors $\frac{3\sqrt{3}}{4}$. Autrement dit : *l'aire d'un triangle isocèle inscrit dans un cercle donné est maximale lorsque le triangle est équilatéral!*
- 6/ L'existence de μ distinct de 0 tel que $f(\mu) = 1$ s'obtient par le fait que f réalise une bijection de $[-1, -\frac{1}{2}]$ sur $[0, \frac{3\sqrt{3}}{4}]$. Elle se conjecture à partir de la représentation.
Essayons de le retrouver avec MAXIMA.

> solve(f(x)=1);

$$\left[x = \frac{\sqrt{1-x^2}-1}{\sqrt{1-x^2}} \right]$$

La réponse est ce qu'elle est, loin d'être satisfaisante. En remarquant que f est positive, l'équation $f(x) = 1$ est équivalente à $f(x)^2 = 1$.

> `M:transpose(solve(f(x)^2=1));`

$$\left(\begin{array}{l} x = \frac{4\left(\frac{\sqrt{3}i}{2} - \frac{1}{2}\right)}{9\left(\frac{\sqrt{11}}{3\sqrt{3}} - \frac{19}{27}\right)^{\frac{1}{3}}} + \left(\frac{\sqrt{11}}{3\sqrt{3}} - \frac{19}{27}\right)^{\frac{1}{3}} \left(-\frac{\sqrt{3}i}{2} - \frac{1}{2}\right) + \frac{2}{3} \\ x = \left(\frac{\sqrt{11}}{3\sqrt{3}} - \frac{19}{27}\right)^{\frac{1}{3}} \left(\frac{\sqrt{3}i}{2} - \frac{1}{2}\right) + \frac{4\left(-\frac{\sqrt{3}i}{2} - \frac{1}{2}\right)}{9\left(\frac{\sqrt{11}}{3\sqrt{3}} - \frac{19}{27}\right)^{\frac{1}{3}}} + \frac{2}{3} \\ x = \left(\frac{\sqrt{11}}{3\sqrt{3}} - \frac{19}{27}\right)^{\frac{1}{3}} + \frac{4}{9\left(\frac{\sqrt{11}}{3\sqrt{3}} - \frac{19}{27}\right)^{\frac{1}{3}}} + \frac{2}{3} \\ x = 0 \end{array} \right)$$

Il y a quatre solutions dont deux sont, selon les apparences, complexes. On retrouve 0 comme racine réelle et μ qui doit être en avant dernière position.

Quelle a été l'équation effectivement résolue ?

> `e:expand(f(x)^2-1=0);`

$$-x^4 + 2x^3 - 2x = 0$$

C'est une équation du quatrième degré...

```
> factor(e);
```

$$-x(x^3 - 2x^2 + 2) = 0$$

Qui, en réalité, se réduit à la résolution d'une équation du troisième degré, ce que MAXIMA sait faire (méthode de CARDAN).

Isolons maintenant une expression de μ :

```
> mu:rhs(part(solve(f(x)^2=1),3));
```

$$\left(\frac{\sqrt{11}}{3\sqrt{3}} - \frac{19}{27}\right)^{\frac{1}{3}} + \frac{4}{9\left(\frac{\sqrt{11}}{3\sqrt{3}} - \frac{19}{27}\right)^{\frac{1}{3}}} + \frac{2}{3}$$

Donnons-en une approximation :

```
> a:float(bfloat(float(mu)));
```

-0.83928675303104

Voici une approximation de μ qui semble satisfaisante.

```
> f(a);
```

1.000000005011567

Enfin, tout est relatif. Les derniers chiffres donnés dans l'approximation de μ ne sont pas significatifs

